NAG Library Routine Document

G01FDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01FDF returns the deviate associated with the given lower tail probability of the F or variance-ratio distribution with real degrees of freedom, via the routine name.

2 Specification

```
FUNCTION G01FDF (P, DF1, DF2, IFAIL)
REAL (KIND=nag_wp) G01FDF
INTEGER            IFAIL
REAL (KIND=nag_wp) P, DF1, DF2
```

3 Description

The deviate, f_p, associated with the lower tail probability, p, of the F-distribution with degrees of freedom ν_1 and ν_2 is defined as the solution to

$$P\left(F \leq f_p : \nu_1, \nu_2\right) = p = \frac{\nu_1^{\frac{1}{2}\nu_1}\nu_2^{\frac{1}{2}\nu_2}\Gamma\left(\frac{\nu_1+\nu_2}{2}\right)}{\Gamma\left(\frac{\nu_1}{2}\right)\Gamma\left(\frac{\nu_2}{2}\right)}\int_0^{f_p} F^{\frac{1}{2}(\nu_1-2)}\left(\nu_2+\nu_1 F\right)^{-\frac{1}{2}(\nu_1+\nu_2)}\,dF,$$

where $\nu_1, \nu_2 > 0$; $0 \leq f_p < \infty$.

The value of f_p is computed by means of a transformation to a beta distribution, $P_\beta(B \leq \beta : a, b)$:

$$P(F \leq f : \nu_1, \nu_2) = P_\beta\left(B \leq \frac{\nu_1 f}{\nu_1 f + \nu_2} : \nu_1/2, \nu_2/2\right)$$

and using a call to G01FEF.

For very large values of both ν_1 and ν_2, greater than 10^5, a normal approximation is used. If only one of ν_1 or ν_2 is greater than 10^5 then a χ^2 approximation is used; see Abramowitz and Stegun (1972).

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

5 Parameters

1: P – REAL (KIND=nag_wp) *Input*

On entry: p, the lower tail probability from the required F-distribution.

Constraint: $0.0 \leq \mathrm{P} < 1.0$.

2: DF1 – REAL (KIND=nag_wp) *Input*

On entry: the degrees of freedom of the numerator variance, ν_1.

Constraint: DF1 > 0.0.

3: DF2 – REAL (KIND=nag_wp) *Input*

On entry: the degrees of freedom of the denominator variance, ν_2.

Constraint: DF2 > 0.0.

4: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G01FDF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

If on exit IFAIL $= 1$, 2 or 4, then G01FDF returns 0.0.

IFAIL $= 1$

On entry, P < 0.0,
or P ≥ 1.0.

IFAIL $= 2$

On entry, DF1 ≤ 0.0,
or DF2 ≤ 0.0.

IFAIL $= 3$

The solution has not converged. The result should still be a reasonable approximation to the solution. Alternatively, G01FEF can be used with a suitable setting of the parameter TOL.

IFAIL $= 4$

The value of P is too close to 0 or 1 for the value of f_p to be computed. This will only occur when the large sample approximations are used.

7 Accuracy

The result should be accurate to five significant digits.

8 Further Comments

For higher accuracy G01FEF can be used along with the transformations given in Section 3.

9 Example

This example reads the lower tail probabilities for several F-distributions, and calculates and prints the corresponding deviates until the end of data is reached.

9.1 Program Text

```
    Program g01fdfe

!     G01FDF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: g01fdf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: df1, df2, f, p
      Integer                          :: ifail
!     .. Executable Statements ..
      Write (nout,*) 'G01FDF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Display titles
      Write (nout,*) '     P      DF1     DF2      F'
      Write (nout,*)

d_lp: Do
        Read (nin,*,Iostat=ifail) p, df1, df2
        If (ifail/=0) Then
          Exit d_lp
        End If

!       Calculate deviates (inverse CDF)
        ifail = -1
        f = g01fdf(p,df1,df2,ifail)
        If (ifail/=0) Then
          If (ifail/=3 .And. ifail/=4) Then
            Exit d_lp
          End If
        End If

!       Display results
        Write (nout,99999) p, df1, df2, f
      End Do d_lp

99999 Format (1X,4F8.3,A,I1)
    End Program g01fdfe
```

9.2 Program Data

```
G01FDF Example Program Data
0.9837  10.0   25.5          :P DF1 DF2
0.9000  1.0    1.0           :P DF1 DF2
0.5342  20.25  1.0           :P DF1 DF2
```

9.3 Program Results

```
 G01FDF Example Program Results

      P       DF1      DF2       F

    0.984  10.000  25.500   2.837
    0.900   1.000   1.000  39.863
    0.534  20.250   1.000   2.500
```